```
type MARK = {subst, subst, nadj, foot, anchor, coanchor, flex}
type CAT = {np,v,vp,s}
type PHON = {e}
type CASE = {nom,acc,none}
type AGR = [ pers : PERS,
             thirdsing : bool,
             num : NUM,
             gen : GEN ]
type NUM = {plur,sing}
type PERS = {1,2,3}
type GEN = {masc,fem,neuter}

property mark : MARK

feature cat : CAT
feature case : CASE
feature agr : AGR

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% TREE FRAGMENTS:
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

class VerbProjection
export  ?VP ?V ?VMARK ?AGR
declare  ?VP ?V ?VMARK ?AGR
{<syn>{
        node ?VP [cat = vp];
        node ?V (mark = ?VMARK)[cat = v, agr = ?AGR];
        ?VP -> ?V
        }
}


%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

class Subject
export ?S ?NP ?VP ?NPMARK ?AGR
declare ?S ?NP ?VP ?NPMARK ?AGR
{ <syn>{
        node ?S [cat = s]{
                node ?NP (mark = ?NPMARK)[cat=np, case = nom, agr = ?AGR]
                node ?VP [cat = vp]
                }
        }
}

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

class Object
export ?VP ?NP ?V ?NPMARK
declare ?VP ?NP ?V ?NPMARK
{ <syn>{
        node ?VP [cat = vp];
        node ?NP (mark=?NPMARK)[cat=np, case = acc];
        node ?V [cat=v];
        ?VP -> ?NP;   ?VP -> ?V; ?V >> ?NP
        }
}

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

class EmptyWord
export ?XP ?CASE ?AGR
declare ?XP ?CASE ?AGR
{ <syn>{
        node ?XP (mark=flex)[phon = e, case = ?CASE, agr = ?AGR]
        }
}

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

class WhNP
export ?S0 ?S2 ?NP ?WHMARK ?WHCASE ?WHAGR
declare ?S0 ?S2 ?NP ?WHMARK ?WHCASE ?WHAGR
{ <syn>{
        node ?S0 [cat = s] {
                node ?NP (mark=?WHMARK)[cat = np, case = ?WHCASE, agr = ?WHAGR]
                node ?S2 [cat = s]
                }
        }
}

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

class nx0V
import VerbProjection[]
export ?S ?NP0 ?NP0MARK
declare ?Subj ?S ?NP0 ?NP0MARK
{
        ?Subj = Subject[];
        ?VP = ?Subj.?VP;
        ?AGR = ?Subj.?AGR;

        ?NP0 = ?Subj.?NP;
        ?S = ?Subj.?S;
        ?NP0MARK = ?Subj.?NPMARK
}

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

class nx0Vnx1
import nx0V[]
export ?NP1 ?NP1MARK
declare ?NP1 ?NP1MARK ?Obj
{
        ?Obj = Object[];
        ?Obj.?VP = ?VP;
        ?Obj.?V = ?V;

        ?NP1 = ?Obj.?NP;
        ?NP1MARK = ?Obj.?NPMARK
}

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

class Wnx0Vnx1
import nx0Vnx1[] WhNP[]
{
        ?S2 = ?S
}

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

class Wnx0V
import nx0V[] WhNP[]
{
        ?S2 = ?S
}


%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% TREE TEMPLATES:
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

class alphanx0V
import nx0V[]
{
        ?NP0MARK = subst;
        ?VMARK = anchor
}

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

class alphanx0Vnx1
import nx0Vnx1[]
{
        ?NP0MARK = subst;
        ?VMARK = anchor;
```

```
        ?NP1MARK = subst
}

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

class alphaW1nx0Vnx1
import Wnx0Vnx1[]
declare ?EmptyW
{
        ?EmptyW = EmptyWord[];
        ?EmptyW.?XP = ?NP1;

        ?EmptyW.?CASE = ?WHCASE;
        ?EmptyW.?AGR = ?WHAGR;

        ?NP0MARK = subst;
        ?VMARK = anchor;
        ?WHMARK = subst
}

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

class alphaW0nx0Vnx1
import Wnx0Vnx1[]
declare ?EmptyW
{
        ?EmptyW = EmptyWord[];
        ?EmptyW.?XP = ?NP0;

        ?EmptyW.?CASE = ?WHCASE;
        ?EmptyW.?AGR = ?WHAGR;

        ?NP1MARK = subst;
        ?VMARK = anchor;
        ?WHMARK = subst
}


%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

class alphaW0nx0V
import Wnx0V[]
declare ?EmptyW
{
        ?EmptyW = EmptyWord[];
        ?EmptyW.?XP = ?NP0;

        ?EmptyW.?CASE = ?WHCASE;
        ?EmptyW.?AGR = ?WHAGR;

        ?VMARK = anchor;
        ?WHMARK = subst
}

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% TREE FAMILIES:
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

class Tnx0V
declare ?Tnx0V
{
        ?Tnx0V = ( alphanx0V[] | alphaW0nx0V[] )
}

class Tnx0Vnx1
declare ?Tnx0V
{
        ?Tnx0V = ( alphanx0Vnx1[] | alphaW0nx0Vnx1[] | alphaW1nx0Vnx1[]  )
}



%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% EVALUTATION:

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

value alphanx0V
value alphanx0Vnx1
value alphaW1nx0Vnx1
value alphaW0nx0Vnx1
value alphaW0nx0V

value Tnx0V
value Tnx0Vnx1
```